Lab 05: Carnot Cycle

Objective

To plot a p-V diagram and T-S diagram for a Carnot cycle using Excel.

Introduction

In this lab, you will use Excel to draw a p-V diagram and T-S diagram for a Carnot cycle. You will also create an animation for the change of the state of the gas and energies for a Carnot engine and Carnot refrigerator.

A heat engine is a device that converts some of the energy it absorbs as heat from a high-temperature reservoir into work and expels the rest of the heat to a low-temperature reservoir. Let us call the temperature of the high-temperature reservoir T_H and the temperature of the low-temperature reservoir T_L. Any engine has a working substance, such as a gasoline-air mixture in a car engine, which must be taken over thermodynamic cycles to produce work on a sustained basis.

A Carnot engine is an ideal heat engine that uses an ideal gas as its working substance. All processes in a Carnot engine are reversible. Compared to any heat engine operating between T_H and T_L, a Carnot engine is the most efficient one.

p-V diagram of a Carnot Engine

We want to plot the p-V diagram for the cycle of a Carnot engine which consists of four strokes as shown in the Fig. 1. In the first stroke, the temperature is kept constant at T_H and the volume of the ideal gas expands from V_a to V_b. The process in which the temperature is kept constant is called an isothermal process. In the second stroke, no heat is transferred to the gas and the gas expands from V_b to V_c and its temperature decreases from T_H to T_L. The process in which no heat is transferred to the gas is called an adiabatic process. In the third stroke, the gas is compressed isothermally at T_L and its volume decreases from V_c to V_d. In the fourth stroke, the gas is compressed adiabatically to its initial state at the beginning of the cycle.



Figure 1. a p-V diagram of a Carnot cycle.

For a reversible adiabatic process:

$$TV^{\gamma-1} = \text{constant.} \quad (1)$$

Here, $\gamma \equiv C_p/C_V$, where C_p is the molar specific heat of an ideal gas at constant pressure and C_V is the molar specific heat of an ideal gas at constant volume. $C_p = C_V + R$. We will use a monatomic gas, for which $C_V = 3R/2$ and $\gamma = 5/3$.

For the first isothermal stroke

$$T = T_H. \tag{2}$$

For the second adiabatic stroke

$$TV^{\gamma-1} = T_H V_b^{\gamma-1} \rightarrow T = T_H (V_b/V)^{\gamma-1}. \tag{3}$$

For the third isothermal stroke

$$T = T_L. \tag{4}$$

For the fourth adiabatic stroke

$$TV^{\gamma-1} = T_H V_a^{\gamma-1} \rightarrow T = T_H (V_a/V)^{\gamma-1}. \tag{5}$$

For any stroke, the pressure p as a function of volume V can be found from the ideal gas law:

$$pV = nRT. \tag{6}$$

Here, n is the number of moles of the gas, and $R = 8.31$ J/mol·K is the gas constant.

Table 1 shows the volume at beginning of each stroke.

Table 1. Volume at the beginning of each stroke.

Quantity	Formula
n	Given
T_H	Given
T_L	Given
V_a	Given
V_b	Given
V_c	$T_H V_b^{\gamma-1} = T_L V_c^{\gamma-1} \rightarrow V_c = V_b \left(\frac{T_H}{T_L}\right)^{1/(\gamma-1)}$
V_d	$T_H V_a^{\gamma-1} = T_L V_d^{\gamma-1} \rightarrow V_d = V_a \left(\frac{T_H}{T_L}\right)^{1/(\gamma-1)}$

Exercise 1:
You will use Excel to plot p-V diagram of a cycle of a Carnot engine.

In the following, you can see detailed instructions for this exercise.

1- Open a new Excel workbook. If column A appears on the right, select Page Layout, then unselect Sheet Right-to-Left



To write γ, write g, select it, then write in the font box Symbol, then press enter.



3- Type the text.

	A	B	C	D	E	F	G	H
1		R		J/mol K	gas constant			
2		Cv		J/mol K	molar specific heat at constant volume			
3		Cp		J/mol K	molar specific heat at constant pressure			
4		γ			gamma = Cp/Cv			
5		n		mole	number of moles			
6		TH		K	temperature of high-temperature reservoir			
7		TL		K	temperature of low-temperature reservoir			
8		Va		m3	Volume at a			
9		Vb		m3	Volume at b			
10		Vc		m3	Volume at c			
11		Vd		m3	Volume at d			

4-Select Page Layout from the main menu, then select Orientation, then select Landscape as shown.



2- change the name of the worksheet by double clicking on its name and change it to Lab-05

1- To change the name of cell C1 to Rg, select cell C1, then click on the name box and change the text from C1 to Rg, then press enter. Now you can use Rg to refer to the content of cell C1.



2-

Name cell C2	Cv.
Name cell C3	Cp.
Name cell C4	gamma.
Name cell C5	n.
Name cell C6	TH.
Name cell C7	TL.
Name cell C8	Va.
Name cell C9	Vb.
Name cell C10	Vc.
Name cell C11	Vd.

4- Decrease or increase number of digits after the decimal point to match the ones shown.



3-

In cell C1, type	8.31.
In cell C2, type	=3*Rg/2.
In cell C3, type	=Cv+Rg.
In cell C4, type	=Cp/Cv.
In cell C5, type	0.001.
In cell C6, type	600.
In cell C7, type	300.
In cell C8, type	0.100.
In cell C9, type	=2.5*Va.
In cell C10, type	=Vb*(TH/TL)^(1/(gamma-1)).
In cell C11, type	=Va*(TH/TL)^(1/(gamma-1))

	B	C	D
1	R	8.31	J/mol K
2	Cv	12.5	J/mol K
3	Cp	20.8	J/mol K
4	γ	1.67	
5	n	0.0010	mole
6	TH	600	K
7	TL	300	K
8	Va	0.100	m3
9	Vb	0.250	m3
10	Vc	0.707	m3
11	Vd	0.283	m3

6- change the widths of column A and B by double clicking at the location of the arrowheads.

5- Type the text.

	A	B	C	D	E
32					
33		Stroke	Volume	Pressure	Temper.
34			V (m3)	p (Pa)	T (K)
35	a				

7- In cell C35, type =Va.
To plot the path on the p-V diagram for stroke $a \rightarrow b$, you will find the pressure value for 20 different volumes equally spaced between V_a and V_b.
In cell C36, type =C35+(Vb-Va)/20.

	A	B	C	D	E
32					
33		Stroke	Volume	Pressure	Temper.
34			V (m3)	p (Pa)	T (K)
35	a		0.100		
36			0.108		

Copy the formula in C36 to the cells bellow till cell C55 by clicking on the right-bottom corner of cell C36, holding, and moving the mouse down to cell C55 then releasing the mouse. You should get the value of Vb at cell C55.
In cell A55, type b.

	A	B	C
54			0.243
55	b		0.250

1- To plot the path on the p-V diagram for stroke $b \rightarrow c$, you will find the pressure value for 20 different volumes equally spaced between Vb and Vc.
In cell C56, type =C55+(Vc-Vb)/20.
Copy the formula in cell C56 to the cells bellow till cell C75. You should get the value of Vc at cell C75.
In cell A75, type c.

	A	B	C
74			0.684
75	c		0.707

2- To plot the path on the p-V diagram for stroke $c \rightarrow d$, you will find the pressure value for 20 different volumes equally spaced between Vc and Vd.
In cell C76, type =C75+(Vd-Vc)/20.
Copy the formula in cell C76 to the cells bellow till cell C95. You should get the value of Vd at cell C95.
In cell A95, type d.

	A	B	C
94			0.304
95	d		0.283

3- To plot the path on the p-V diagram for stroke $d \rightarrow a$, you will find the pressure value for 20 different volumes equally spaced between Vd and Va.
In cell C96, type =C95+(Va-Vd)/20.
Copy the formula in cell C96 to the cells bellow till cell C155. You should get the value of Va at cell C155.
In cell A115, type a.

	A	B	C
114			0.109
115	a		0.100

4-
In cell B35, type 1. Copy it to the cells below it till cell B54 by clicking on the right-bottom corner of cell B35, holding, and moving the mouse down to cell B54 then releasing the mouse.
In cell B55, type 2 and copy it to the cells below it till cell B74.
In cell B75, type 3 and copy it to the cells below it till cell B94.
In cell B95, type 4 and copy it to the cells below it till cell B115.

5-
In cell E35, type =TH.
Copy the formula in cell E35 to the cells below it till cell E55.

	E
54	600
55	600

In cell E56, type =TH*(Vb/C56)^(gamma-1)
Copy the formula in cell E56 to the cells below it till cell E75.

	E
73	314
74	307
75	300

In cell E76, type =TL.
Copy the formula in cell E76 to the cells below it till cell E95.

	E
93	300
94	300
95	300

In cell E96, type =TH*(Va/C96)^(gamma-1).
Copy the formula in cell E96 to the cells below it till cell E115.

	E
113	536
114	566
115	600

1- In cell D35, type =n*Rg*E35/C35.

Copy the formula in cell D35 to the cells below it till cell D115.

	A	B	C	D	E
112		4	0.127	33.3	510
113		4	0.118	37.7	536
114		4	0.109	43.1	566
115	a	4	0.100	49.9	600

2- Click on cell D115. Scroll up until you see cell C35 but do not select it. While pressing shift key, click on cell C35. This will select all the cells in the range between C35 and D115.

	A	B	C	D	E
31					
32					
33		Stroke	Volume	Pressure	Temper.
34			V (m3)	p (Pa)	T (K)
35	a	1	0.100	49.9	600
36		1	0.108	46.4	600

3- Select Insert from the main menu, then from the chart group select scatter with Smooth Line as shown.





4- Move and resize the chart as shown.



5- Select Chart Title and then click delete key.

6- Click on your plot, then click on the plus sign appearing near the top right corner of the plot, then select Axis Titles as shown.



Select the Axis Title of the horizontal axis and change it to Volume (m^3) and select the Axis Title of the vertical axis and change it to Pressure (Pa).



To make 3 superscript, select it then press Ctrl key and 1 key at the same time to open the font dialog box. Select Superscript.



1- To change the decimal places, double click on any number of the horizontal axis to show the Format Axis menu on the right of the worksheet.



In the bottom of the Format Axis menu, open Number tab and change the Decimal places to 1.



Change the decimal places for the vertical axis to 0.

2- move the mouse over the curve and right-click on it. From the menu choose Select Data.



3- Click on Edit.
Change the series name to Carnot Cycle. Then press OK.



Click on Add.



Type a for the Series name.
Select cell C35 for the Series X values.
Select cell D35 for the Series Y values. Then press OK.



Add another series with the name b and select cell C55 for the x values and select cell D55 for the Y values.

Add another series with the name c and select cell C75 for the x values and select cell D75 for the Y values.

Add another series with the name d and select cell C95 for the x values and select cell D95 for the Y values. Then press OK.

1- Double click on your plot to show the Format menu on the right of the worksheet. Click on the arrow on the right of Chart Options and select Series “Carnot Cycle” as shown.



Change the color of the Carnot Cycle curve to gray as shown.



2- Select Series “a” as shown.



Click on Marker.
Click Marker Options.
Select Built in.
Choose circle as shown.
Change the size to 3.



1- Click on the circle representing a, then click on the plus sign appearing near the top right corner of the plot, then select Data Labels, Select Right as shown.



Double click on the label and select [Series] a as shown.



2- Do for series b, c, and d what you did for series a.

You need to change the fill and border color to orange as shown.

Make the Data Labels above series c and to the left of series d.



3- Click on any empty space in your plot.

Select Insert from the main menu, then click on Shapes from Illustrations group, then select Text Box as shown.



Type Carnot Cycle in the text box, change the font size to 14, and move the text box as shown.



Gas state of a Carnot engine

The state of an ideal gas is determined by its volume, pressure, and temperature. If the number of moles does not change, as in the case of a Carnot engine, then the state of the gas is determined only by its pressure and volume. For a Carnot engine, the state of the gas changes in a clockwise manner over the Carnot cycle. We will assume the cycle starts at state a of Fig. 1.

Exercise 2:

You will use Excel to make animation for the state of the ideal gas in the p-V diagram as it changes over the cycle of a Carnot engine. You will use a red dot to indicate the current state of the gas.

In the following, you can see detailed instructions for this exercise.

1-

In cell B28, type row number.
In cell B29, type =B33.
In cell C29, type =C34.
In cell D29, type =D34.
In cell E29, type =E34.



2- To change the name of cell D28 to rn, select cell D28, then click on the name box and change the text from D28 to rn, then press enter. Now you can use rn to refer to the content of cell D28.



3- You want to put in row 30 the values of the next row between row 35 and row 115 every time Excel updates the calculations in all cells.
To do this, you need to set the maximum iterations to 1. Select file from the main menu, then select Options, the select Formulas. Select Enable Iterative Calculation box and Change Maximum iterations to 1.







4-

Condition Value if true Value if false

In cell D28, type =IF(rn<35,35,IF(rn>114,35,rn+1)).
In cell B30, type =INDIRECT("B"&rn).
In cell C30, type =INDIRECT("C"&rn).
In cell D30, type =INDIRECT("D"&rn).
In cell E30, type =INDIRECT("E"&rn).

INDIRECT("B"&rn) gives the value of the cell with address B&rn. For example, if rn =40, then this is the value of cell B40.

Change the decimal places for row 30 to match those of row 35.

5- Change the font color for cell B28 and cell D28 to red.



6- move the mouse over the curve of your plot and right-click on it. From the menu choose Select Data.

Add another series. name it state. Select cell C30 for its x values and select cell D30 for its Y values.



1- Double click on your plot to show the Format menu on the right of the worksheet. Click on the arrow on the right of Chart Options and select Series “state” as shown.



2-
Click on Marker.
Click Marker Options.
Select Built in.
Choose circle as shown.



3- Change the fill and border color to red as shown.



Press F9 key repeatedly and observe the red dot representing the state of the gas moving clockwise.



row number			44	
Stroke		V (m3)	p (Pa)	T (K)
	1	0.168	29.8	600

T-S diagram of a Carnot engine

The entropy of a system S is a state property that does not depend on path. For an ideal gas, the change in entropy ΔS is

$$\Delta S = S_f - S_i = nR \ln \frac{V_f}{V_i} + nC_V \ln \frac{T_f}{T_i}$$

You will plot temperature T as a function of change of entropy with respect of state a: $\Delta S = S - S_a$ for a cycle of a Carnot engine.

Exercise 3:
You will use Excel to plot a T-S diagram a cycle of a Carnot engine.

In the following, you can see detailed instructions for this exercise.

1- Type the text.

	E	F
33	Temper.	Chagne in Entory
34	T (K)	(S-Sa) (mJ/K)

Note the unit is **milli** Joule per Kelvin (mJ/K).

Adjust the width of F column by double clicking on the line between letter F and letter G in the column titles.

2-
In cell F35, type =1000*(n*Rg*LN(C35/Va)+n*Cv*LN(E35/TH)).

Note we multiply by 1000 because our unit is in **milli**.

Copy the formula in F35 to the cells bellow till cell F115 by clicking on the right-bottom corner of cell F35, holding, and moving the mouse down to cell F115 then releasing the mouse.

In cell F29, type =F34.
In cell F30, type =INDIRECT("F"&rn).

Adjust the decimal places to 2.

	F	G
29	(S-Sa) (mJ/K)	
30	5.00	
31		
32		
33	Chagne in Entory	
34	(S-Sa) (mJ/K)	
35	0.00	
36	0.60	
37	1.16	

3- Right click on the chart area but outside the plot area and select copy as shown.



Select any empty cell and then press Ctrl key and v key at the same time to paste the copied chart.

Move the pasted chart just to the right of the original chart.

Change the axis title for the horizontal axis to Change in Entropy (S-Sa) (mJ/K).

Select the letter a of Sa, then press Ctrl key and 1 key at the same time. Then select subscript.

Change the axis title for the vertical axis to Temperature T (K).

1- Right-click on the pasted chart and select from the menu Select Data….

Select Series Carnot Cycle, then select Edit.





In the X values change letter C with F.
In the Y values change letter D with E.

Do the same for
Series a
Series b
Series d
Series c
Series state

2- As you did for the P-V diagram:

Change the color of series Carnot Cycle to gray.

For series a, b, c, and d, select built-in marker circle, make its size 3, and make its color orange.

For Series state, select built-in marker circle, make its size 5, and make its color red.

Add data label for series a, b, c, and d. For a and b make them above the data point, while for c and d make them below the data point.

3- Change the range for the horizontal axis from 0 to 8 and for the vertical axis from 200 to 700. Make Units Major 100 for the vertical axis.



Press F9 key repeatedly and observe the red dot representing the state of the gas moving clockwise for both diagrams.



Work and heat for a Carnot engine

Work W and heat Q are not state properties; they depend on path. In the following, you will find the accumulated work and heat as the gas state follows a Carnot cycle starting at state a of Fig. 1 and just for one cycle. You will get different values if you start with another state or go over more than one cycle. The internal energy E_{int} of a system is a state property that does not depend on path, and the change in the internal energy ΔE_{int}, according to the first law of thermodynamics, is

$$\Delta E_{int} = Q - W. \tag{7}$$

For an ideal gas $E_{int} = nC_V T$.

For an isothermal process, for which the temperature is constant,

$$\left.\begin{array}{c} T = \text{constant} \\ E_{int} = nC_V T = \text{constant} \\ Q = W = nRT \ln \dfrac{V_f}{V_i} \end{array}\right\} \quad \text{isothermal process.}$$

For an adiabatic process, for which no heat is transferred to the gas $Q = 0$,

$$\left.\begin{array}{c} \Delta E_{int} = nC_V \Delta T \\ Q = 0 \\ W = -nC_V \Delta T \end{array}\right\} \quad \text{adiabatic process.}$$

The table below shows how to calculate the accumulated work starting at state a for one cycle of a Carnot engine. The heat starting from state a can be obtained from the first law of thremodynamcis, Eq. 7.

Table 2. Work starting at point a to a state on the cycle of a Carnot engine for the first cycle.

On stroke	Work done by the gas in the first cycle starting from state a
$a \to b$	$nRT_H \ln \dfrac{V}{V_a}$
$b \to c$	$W_{a\to b,Eng} - nC_V(T - T_H)$
$c \to d$	$W_{a\to c,Eng} + nRT_L \ln \dfrac{V}{V_c}$
$d \to a$	$W_{a\to d,Eng} - nC_V(T - T_L)$

$$W_{a\to b,Eng} = nRT_H \ln \frac{V_b}{V_a}$$
$$W_{a\to c,Eng} = W_{a\to b,Eng} - nC_V(T_L - T_H)$$
$$W_{a\to d,Eng} = W_{a\to c,Eng} + nRT_L \ln \frac{V_d}{V_c}$$

Exercise 4:

You will use Excel to calculate the internal energy E_{int} at any state of the Carnot cycle. Also, you will calculate the work W done the gas and heat Q absorbed by the gas starting at point a for the first cycle. You will use a chart plot the show ΔE_{int}, W, and Q as the state of the gas changes over the cycle of a Carnot engine.

In the following, you can see detailed instructions for this exercise.

1- Select cell H31 to K31 and merge them together.
In the merged cell, type Starting at point a for one cycle.
Type the text.

	G	H	I	J	K
31		Starting at point a for one cycle			
32		Engine	Engine	Refriger.	Refriger.
33	Internal Energy	Work	Heat	Work	Heat
34	Eint (J)	W (J)	Q (J)	W (J)	Q (J)
35					

2-
Name cell G35 Einta
Name cell H55 WabEng.
Name cell H75 WacEng.
Name cell H95 WadEng.

3- In cell G35, type =n*CV*E35
Change the decimal places to 2 for cell G35.
Copy the formula in G35 to the cells bellow till cell G115.

4-
In cell H35, type =n*Rg*TH*LN(C35/Va)
Copy the formula in H35 to the cells bellow till cell H55.
Change the decimal places to 2 for the range H35 to H55.

In cell H56, type =WabEng-n*CV*(E56-TH)
Copy the formula in H56 to the cells bellow till cell H75.

In cell H76, type =WacEng+n*Rg*TL*LN(C76/Vc)
Copy the formula in H76 to the cells bellow till cell H95.

In cell H96, type =WadEng-n*CV*(E96-TL)
Copy the formula in H96 to the cells bellow till cell H115.

4-
In cell I35, type =(G35-Einta)+H35
Change the decimal places to 2 for cell I35.

Copy the formula in I35 to the cells bellow till cell I115.

5- Type the text

	H
28	ΔEint (J)
29	W (J)
30	Q (J)

6-
In cell I28, type =INDIRECT("G"&rn)-Einta.
Change the decimal places to 2 for cell I28.
In cell I29, type =INDIRECT("H"&rn).
Change the decimal places to 2 for cell I29.
In cell I30, type =INDIRECT("I"&rn).
Change the decimal places to 2 for cell I30.

7- Select the range as shown

	G	H	I
27			
28		ΔEint (J)	-0.792
29		W (J)	5.36
30		Q (J)	4.57
31		Starting at point a for	

1- Select Insert from the main menu, then from the chart group select Insert Column or Bar Chart as shown.





Change Chart Title to Energy
Change the range for the vertical axis to -10 to 10 and its decimal places to 0.



2- Move and resize the chart as shown.



Press F9 key repeatedly and observe how ΔEint, W, and Q change over the cycle of a Carnot engine .

Carnot Refrigerator

A refrigerator is a device that uses work to transfer heat from a low-temperature reservoir to a high-temperature reservoir. A Carnot refrigerator is an ideal refrigerator that uses an ideal gas as its working substance. The gas state follows a cycle of a Carnot engine but in counterclockwise manner.

The table below shows you how to calculate the accumulated work starting at state a for one cycle of a Carnot refrigerator.

Table 3. Work starting at point a to a state on the cycle of a Carnot refrigerator for the first cycle.

On stroke	Work done by the gas in the first cycle starting from state a
$a \to d$	$-nC_V(T - T_H)$
$d \to c$	$W_{a\to d,Ref} + nRT_L \ln\frac{V}{V_d}$
$c \to b$	$W_{a\to c,Ref} - nC_V(T - T_L)$
$b \to a$	$W_{a\to b,Ref} + nRT_H \ln\frac{V}{V_b}$

$$W_{a\to d,Ref} = -nC_V(T_L - T_H)$$
$$W_{a\to c,Ref} = W_{a\to d,Ref} + nRT_L \ln\frac{V_c}{V_d}$$
$$W_{a\to b,Ref} = W_{a\to c,Ref} - nC_V(T_H - T_L)$$

Exercise 5:

You will use Excel to calculate the work W done by the gas and heat Q absorbed by the gas starting at point a for the first cycle of a Carnot refrigerator. Also, you will modify your Excel worksheet so that you can choose an engine or refrigerator in the animations you did previously.

In the following, you can see detailed instructions for this exercise.

1-
Name cell J55 WabRef.
Name cell J75 WacRef.
Name cell J95 WadRef.

2-
In cell J35, type =WabRef+n*Rg*TH*LN(C35/Vb)
Copy the formula in J35 to the cells bellow till cell J54.
Change the decimal places to 2 for the range J35 to J54.

In cell J55, type =WacRef-n*Cv*(E55-TL)
Copy the formula in J55 to the cells bellow till cell J74.

In cell J75, type =WadRef+n*Rg*TL*LN(C75/Vd)
Copy the formula in J75 to the cells bellow till cell J94.

In cell J95, type =-n*Cv*(E95-TH)
Copy the formula in J95 to the cells bellow till cell J115.

3-
In cell K35, type =(G35-Einta)+J35
Change the decimal places to 2 for cell K35.

Copy the formula in K35 to the cells bellow till cell K115.

4- You will use the text in cell K28 to indicate whether to operate the cycle as an engine or refrigerator. If the text is “eng” then it is an engine and if it is “ref” then it is a refrigerator.
Name cell K28 sel
In cell K28, type eng
Change the color of cell K28 to red.



5- Cell D28 is the row number which you use to indicate the state.
Change it from
=IF(rn<35,35,IF(rn>114,35,rn+1))
To
=SWITCH(sel,"ref",IF(rn<35,115,IF(rn=35,115,rn-1)),"eng",IF(rn<35,35,IF(rn>114,35,rn+1)),35)

Value to switch
Value to match1
Value to return for match1
Value to match2
Value to return for match2
Value to return for no match

6- Change cell I29 from
=INDIRECT("H"&rn)
To
=IF(sel="ref",INDIRECT("J"&rn),INDIRECT("H"&rn)).

7- Change cell I30 from
=INDIRECT("I"&rn)
To
=IF(sel="ref",INDIRECT("K"&rn),INDIRECT("I"&rn)).

8- Change the text in K28 to ref and press F9 key repeatedly and observe P-V diagram, T-S diagram, and energy bar chart.